QT += core gui widgets

CONFIG += c++11 console

DESTDIR = bin

HEADERS += \
    lv_conf.h

SOURCES += \
    lv_main.c \
    main.cpp

include($$PWD/../../src/src.pri)
include($$PWD/app/app.pri)

DEFINES +=  \
    LV_QT_WORKING_DIR=\\\"$$OUT_PWD/$$DESTDIR\\\"

run_image.commands = python $$PWD/../../scripts/image2bin.py -i $$PWD/res -o $$DESTDIR/res
run_image.target = run_image
QMAKE_EXTRA_TARGETS += run_image
PRE_TARGETDEPS += run_image

copy_font.commands = xcopy /Y $$shell_path($$PWD/res/font/bin/*.bin) $$shell_path($$DESTDIR/res/font/)
copy_font.target = copy_font
QMAKE_EXTRA_TARGETS += copy_font
PRE_TARGETDEPS += copy_font
